

# The Primitive Euclidean Algorithm and Intermediate Expression Swell.

Michael Monagan, Spring 2021.

```
> restart;
> a := 3*x^3+x^2+x+5;

$$a := 3x^3 + x^2 + x + 5 \quad (1)$$

```

```
> b := 5*x^2-3*x+1;

$$b := 5x^2 - 3x + 1 \quad (2)$$

```

Pseudo division in  $\mathbb{Z}[x]$ : prem( a, b, x ) = rem( lc(b)^(deg(a) - deg(b) + 1), a, b, x ) = rem( 5^2 a, b, x )

```
> prem(a,b,x);

$$52x + 111 \quad (3)$$

```

```
> rem(5^2*a,b,x);

$$52x + 111 \quad (4)$$

```

```
> pr := prem(a,b,x,'m','pq');

$$pr := 52x + 111 \quad (5)$$

```

```
> m, pq;

$$25, 15x + 14 \quad (6)$$

```

```
> expand( m*a=b*pq+pr );

$$75x^3 + 25x^2 + 25x + 125 = 75x^3 + 25x^2 + 25x + 125 \quad (7)$$

```

An example of pseudo-division in  $\mathbb{Z}[y][x]$

```
> a := 5*x^3+y*x^2+3;

$$a := 5x^3 + yx^2 + 3 \quad (8)$$

```

```
> b := 3*y^2*x^2 + 5*x + 7*y;

$$b := 3y^2x^2 + 5x + 7y \quad (9)$$

```

In this example the multiplier  $m = lc(a)^2 = 3^2 \cdot y^4$

```
> pr := prem(a,b,x,'m');

$$pr := -120xy^3 + 6y^4 + 125x + 175y \quad (10)$$

```

```
> m;

$$9y^4 \quad (11)$$

```

```
> rem(m*a,b,x);

$$(-120y^3 + 125)x + 6y^4 + 175y \quad (12)$$

```

## The Primitive Euclidean Algorithm

```
> g := 2*x-3;

$$g := 2x - 3 \quad (13)$$

```

```
> a := expand( g*randpoly(x,degree=4,dense) );

$$a := -14x^5 + 65x^4 - 176x^3 - 23x^2 + 456x - 261 \quad (14)$$

```

```
> b := expand( g*randpoly(x,degree=4,dense) );

$$b := -112x^5 + 168x^4 - 124x^3 + 380x^2 - 437x + 219 \quad (15)$$

```

```
> content(a,x);

$$1 \quad (16)$$

```

```
> content(b,x);

$$1 \quad (17)$$

```

```
> r[0] := primpart(a,x);
```

```

r[1] := primpart(b,x);
k := 1;
while r[k] <> 0 do
  pr := prem(r[k-1],r[k],x);
  r[k+1] := primpart(pr,x);
  k := k+1;
od;
g := r[k-1];
g := gcd(content(a,x),content(b,x))*g; # attach gcd of contents

```

$$\begin{aligned}
r_0 &:= -14x^5 + 65x^4 - 176x^3 - 23x^2 + 456x - 261 \\
r_1 &:= -112x^5 + 168x^4 - 124x^3 + 380x^2 - 437x + 219 \\
&\quad k := 1 \\
pr &:= -4928x^4 + 17976x^3 + 7896x^2 - 57190x + 32298 \\
r_2 &:= -352x^4 + 1284x^3 + 564x^2 - 4085x + 2307 \\
&\quad k := 2 \\
pr &:= -146318080x^3 + 160375552x^2 + 200787904x - 168203328 \\
r_3 &:= -2286220x^3 + 2505868x^2 + 3137311x - 2628177 \\
&\quad k := 3 \\
pr &:= 5568818946161152x^2 - 12794178883681536x + 6661425696659712 \\
r_4 &:= 179778504202x^2 - 413035217061x + 215051191137 \\
&\quad k := 4 \\
pr &:= -14164158620635468468339428800x + 21246237930953202702509143200 \\
r_5 &:= -2x + 3 \\
&\quad k := 5 \\
pr &:= 0 \\
r_6 &:= 0 \\
&\quad k := 6 \\
g &:= -2x + 3 \\
g &:= -2x + 3 \tag{18}
\end{aligned}$$

We need to multiply g by -1 to make it unit normal in  $\mathbb{Z}[x]$ .

Notice that the degree of the pseudo-remainder pr is decreasing until we get pr = gcd BUT the size of the integer coefficients are growing. Let us spy on the size of the integer coefficients for a larger example.

```

> a := expand( g*randpoly(x,degree=20,dense) );
b := expand( g*randpoly(x,degree=20,dense) );
a := 8x21 + 154x20 - 229x19 - 154x18 + 350x17 - 406x16 + 328x15 - 274x14
  + 247x13 + 99x12 - 205x11 - 16x10 + 59x9 - 204x8 + 226x7 - 196x6 - 81x5
  + 409x4 - 288x3 - 130x2 + 78x + 216
b := -74x21 + 157x20 - 243x19 + 173x18 + 74x17 - 109x16 + 340x15 - 89x14 \tag{19}

```

$$+ 152x^{13} - 167x^{12} + 34x^{11} - 277x^{10} + 263x^9 + 131x^8 - 53x^7 - 221x^6 + 282x^5 \\ - 425x^4 + 137x^3 + 224x^2 - 92x + 93$$

```
> r[0] := primpart(a,x);
r[1] := primpart(b,x);
k := 1;
while r[k] <> 0 do
    printf("k=%2d deg=%2d size=%d\n",
           k,degree(r[k],x),length(maxnorm(r[k])));
    pr := prem(r[k-1],r[k],x);
    r[k+1] := primpart(pr,x);
    k := k+1;
od;
g := r[k-1];
g := gcd(content(a,x),content(b,x))*g; # attach gcd of contents
k= 1 deg=21 size=3
k= 2 deg=20 size=5
k= 3 deg=19 size=9
k= 4 deg=18 size=13
k= 5 deg=17 size=17
k= 6 deg=16 size=21
k= 7 deg=15 size=25
k= 8 deg=14 size=29
k= 9 deg=13 size=33
k=10 deg=12 size=39
k=11 deg=11 size=43
k=12 deg=10 size=48
k=13 deg= 9 size=53
k=14 deg= 8 size=57
k=15 deg= 7 size=61
k=16 deg= 6 size=65
k=17 deg= 5 size=71
k=18 deg= 4 size=75
k=19 deg= 3 size=79
k=20 deg= 2 size=84
k=21 deg= 1 size=1
```

$$g := 2x - 3 \\ g := 2x - 3 \quad (20)$$

Notice that the size of the coefficients increases by just over 4 digits each iteration. This is a linear growth in the size of the coefficients. The degree is dropping but coefficients grow until we get to the gcd.

I'm going to redo the example with two variables instead of one so for  $\mathbb{Z}[x,y]$  and we are going to watch what happens to the degree of the pseudo-remainders in  $y$ .

```
> g := 2*x-3*y-5;
g := 2x - 3y - 5 \quad (21)
```

```
> a := expand( g*randpoly([x,y],degree=10,dense) );
a := -102x^{11} + 307x^{10}y - 229x^9y^2 - 59x^8y^3 + 54x^7y^4 + 139x^6y^5 - 237x^5y^6 \\ + 286x^4y^7 - 251x^3y^8 + 253x^2y^9 - 388xy^{10} + 141y^{11} + 445x^{10} - 668x^9y \\ + 24x^8y^2 - 26x^7y^3 + 72x^6y^4 + 141x^5y^5 + 113x^4y^6 - 279x^3y^7 - 219x^2y^8 \\ - 376xy^9 + 352y^{10} - 365x^9 - 110x^8y - 362x^7y^2 + 669x^6y^3 + 265x^5y^4 - 15x^4y^5 \\ - 255x^3y^6 - 293x^2y^7 + 22xy^8 + 354y^9 - 329x^8 + 75x^7y + 448x^6y^2 - 481x^5y^3
```

$$\begin{aligned}
& + 143x^4y^4 - 432x^3y^5 - 380x^2y^6 - 194xy^7 + 481y^8 - 39x^7 - 281x^6y - 161x^5y^2 \\
& - 184x^4y^3 + 327x^3y^4 - 433x^2y^5 + 436xy^6 + 651y^7 + 259x^6 + 699x^5y - 201x^4y^2 \\
& + 515x^3y^3 - 497x^2y^4 + 468xy^5 + 386y^6 + 276x^5 + 492x^4y + 171x^3y^2 - 16x^2y^3 \\
& + 229xy^4 - 195y^5 + 314x^4 - 164x^3y - 310x^2y^2 - 53xy^3 - 71y^4 + 240x^3 \\
& - 459x^2y - 392xy^2 + 232y^3 + 142x^2 - 73xy + 250y^2 - 331x - 361y - 60
\end{aligned}$$

```

> b := expand( g*randpoly([x,y],degree=9,dense) );
b := -50x10 - 117x9y + 168x8y2 + 2x7y3 + 147x6y4 + 2x5y5 + 201x4y6 + 67x3y7      (23)
      + 202x2y8 - 159xy9 - 108y10 + 225x9 + 246x8y + 286x7y2 + 687x6y3 + 98x5y4
      + 718x4y5 + 220x3y6 - 95x2y7 - 206xy8 - 453y9 - 236x8 + 257x7y + 352x6y2
      - 98x5y3 + 108x4y4 - 138x3y5 - 108x2y6 - 19xy7 - 389y8 - 171x7 - 6x6y
      - 40x5y2 - 391x4y3 - 208x3y4 + 353x2y5 + 347xy6 - 43y7 + 332x6 + 168x5y
      - 424x4y2 - 641x3y3 + 159x2y4 + 526xy5 - 174y6 - 64x5 - 142x4y + 216x3y2
      + 12x2y3 + 363xy4 - 15y5 + 388x4 - 15x3y - 333x2y2 - 150xy3 - 430y4
      - 365x3 + 86x2y - 236xy2 - 27y3 - 96x2 - 340xy + 596y2 - 454x + 526y
      + 485

> r[0] := primpart(a,x):
r[1] := primpart(b,x):
k := 1:
while r[k] <> 0 do
    printf("k=%2d degx=%2d degy=%3d size=%d\n",
           k,degree(r[k],x),degree(r[k],y),length(maxnorm(r[k])))
;
    pr := prem(r[k-1],r[k],x);
    r[k+1] := primpart(pr,x);
    k := k+1;
od:
g := r[k-1];
g := gcd(content(a,x),content(b,x))*g; # attach gcd of con
k= 1 degx=10 degy= 10 size=3
k= 2 degx= 9 degy= 11 size=7
k= 3 degx= 8 degy= 14 size=12
k= 4 degx= 7 degy= 19 size=16
k= 5 degx= 6 degy= 26 size=21
k= 6 degx= 5 degy= 35 size=26
k= 7 degx= 4 degy= 46 size=31
k= 8 degx= 3 degy= 59 size=36
k= 9 degx= 2 degy= 74 size=41
k=10 degx= 1 degy= 1 size=1

```

$$\begin{aligned}
g &:= 2x - 3y - 5 \\
g &:= 2x - 3y - 5
\end{aligned} \tag{24}$$

There is now a two dimensional linear growth. The degree in  $y$  is growing and the size of the integer coefficients is also growing. This means the intermediate pseudo-remainders, before we get to the gcd, can be much larger than the input polynomials. Let us make the example larger and time Maple executing the primitive Euclidean algorithm and then time Maple doing the gcd using whatever algorithm it uses.

```
> a := expand( g*randpoly([x,y],degree=40,dense) );
```

```

> b := expand( g*randpoly([x,y],degree=39,dense) ):
> st := time():
r[0] := primpart(a,x):
r[1] := primpart(b,x):
k := 1:
while r[k] <> 0 do
    printf("k=%2d degx=%2d degy=%3d size=%d\n",
           k,degree(r[k],x),degree(r[k],y),length(maxnorm(r[k])) )
;
    pr := prem(r[k-1],r[k],x);
    r[k+1] := primpart(pr,x);
    k := k+1;
od:
g := r[k-1];
g := gcd(content(a,x),content(b,x))*g; # attach gcd of contents
time=time()-st;

```

k= 1 degx=40 degy= 40 size=3  
k= 2 degx=39 degy= 41 size=7  
k= 3 degx=38 degy= 44 size=12  
k= 4 degx=37 degy= 49 size=16  
k= 5 degx=36 degy= 56 size=21  
k= 6 degx=35 degy= 65 size=26  
k= 7 degx=34 degy= 76 size=32  
k= 8 degx=33 degy= 89 size=37  
k= 9 degx=32 degy=104 size=42  
k=10 degx=31 degy=121 size=48  
k=11 degx=30 degy=140 size=54  
k=12 degx=29 degy=161 size=59  
k=13 degx=28 degy=184 size=65  
k=14 degx=27 degy=209 size=71  
k=15 degx=26 degy=236 size=76  
k=16 degx=25 degy=265 size=82  
k=17 degx=24 degy=296 size=88  
k=18 degx=23 degy=329 size=94  
k=19 degx=22 degy=364 size=101  
k=20 degx=21 degy=401 size=107  
k=21 degx=20 degy=440 size=113  
k=22 degx=19 degy=481 size=119  
k=23 degx=18 degy=524 size=125  
k=24 degx=17 degy=569 size=132  
k=25 degx=16 degy=616 size=137  
k=26 degx=15 degy=665 size=144  
k=27 degx=14 degy=716 size=150  
k=28 degx=13 degy=769 size=156  
k=29 degx=12 degy=824 size=163  
k=30 degx=11 degy=881 size=169  
k=31 degx=10 degy=940 size=176  
k=32 degx= 9 degy=1001 size=182  
k=33 degx= 8 degy=1064 size=188  
k=34 degx= 7 degy=1129 size=195  
k=35 degx= 6 degy=1196 size=202  
k=36 degx= 5 degy=1265 size=208  
k=37 degx= 4 degy=1336 size=214  
k=38 degx= 3 degy=1409 size=220  
k=39 degx= 2 degy=1484 size=226  
k=40 degx= 1 degy= 1 size=1

$$g := -2x + 3y + 5$$

$$g := -2x + 3y + 5$$

time = 346.809

(25)

```
> st := time(): gcd(a,b); time()-st;
2 x - 3 y - 5
0.024
```

(26)

Maple is NOT using the primitive Euclidean algorithm! If we added a third variable z to this example, Maple might never finish. Basically, the complexity of the primitive Euclidean algorithm is exponential in the number of variables.

In the 1970s polynomial gcd computation was a major area of research because it was impossible to compute the gcd of two polynomials in many variables for even small degrees. This phenomenon where the intermediate expressions are much larger than the inputs and output(s) has been termed *intermediate expression swell*. It also occurs in Gaussian elimination when applied to a matrix of integers or polynomials.